Linear perturbation of the metric tensor

Preliminaries

Load tensor package

In[2]:=
```
<< xAct`xPert`
```

```
------------------------------------------------------------
Package xAct`xPerm`  version 1.2.3, {2015, 8, 23}
CopyRight (C) 2003-2020, Jose M. Martin-Garcia, under the General Public License.
Connecting to external linux executable...
Connection established.
------------------------------------------------------------
Package xAct`xTensor`  version 1.1.4, {2020, 2, 16}
CopyRight (C) 2002-2020, Jose M. Martin-Garcia, under the General Public License.
------------------------------------------------------------
Package xAct`xPert`  version 1.0.6, {2018, 2, 28}
CopyRight (C) 2005-2018, David Brizuela, Jose M. Martin-Garcia
  and Guillermo A. Mena Marugan, under the General Public License.
------------------------------------------------------------
These packages come with ABSOLUTELY NO WARRANTY; for details type
  Disclaimer[]. This is free software, and you are welcome to redistribute
  it under certain conditions. See the General Public License for details.
------------------------------------------------------------
** Variable $PrePrint assigned value ScreenDollarIndices
** Variable $CovDFormat changed from Prefix to Postfix
** Option AllowUpperDerivatives of ContractMetric changed from False to True
** Option MetricOn of MakeRule changed from None to All
** Option ContractMetrics of MakeRule changed from False to True
```

Nicer printing

In[3]:=
```
$PrePrint = ScreenDollarIndices;
```

In[4]:=
```
$CovDFormat = "Prefix";
```

Object definitions

Spacetime manifold

The spacetime manifold M, on which tensors will be defined. Some Greek letters are defined as tangent space indices.

In[5]:=
```
DefManifold[M, 4, {α, β, γ, μ, ν, ρ, σ, τ, κ, ξ, ζ}]
** DefManifold : Defining manifold M.
** DefVBundle : Defining vbundle TangentM .
```

Metric

The metric g of signature (-,+,+,+). The Levi-Civita derivative of a tensor A_μ will be written as $\nabla_\mu A_\nu$ in prefix notation or $A_{\nu;\mu}$ in postfix notation.

In[6]:=
```
DefMetric[{3, 1, 0}, Met[-μ, -ν], CD, {";", "▿"}, PrintAs → "g"]
** DefTensor : Defining symmetric metric tensor Met[-μ, -ν].
** DefTensor : Defining antisymmetric tensor epsilonMet [-α, -β, -γ, -ζ].
** DefTensor : Defining tetrametric TetraMet [-α, -β, -γ, -ζ].
** DefTensor : Defining tetrametric TetraMet† [-α, -β, -γ, -ζ].
** DefCovD : Defining covariant derivative CD[-μ].
** DefTensor : Defining vanishing torsion tensor TorsionCD [α, -β, -γ].
** DefTensor : Defining symmetric Christoffel tensor ChristoffelCD [α, -β, -γ].
** DefTensor : Defining Riemann tensor RiemannCD [-α, -β, -γ, -ζ].
** DefTensor : Defining symmetric Ricci tensor RicciCD [-α, -β].
** DefCovD : Contractions of Riemann automatically replaced by Ricci .
** DefTensor : Defining Ricci scalar RicciScalarCD [].
** DefCovD : Contractions of Ricci automatically replaced by RicciScalar .
** DefTensor : Defining symmetric Einstein tensor EinsteinCD [-α, -β].
** DefTensor : Defining Weyl tensor WeylCD [-α, -β, -γ, -ζ].
** DefTensor : Defining symmetric TFRicci tensor TFRicciCD [-α, -β].
** DefTensor : Defining Kretschmann scalar KretschmannCD [].
** DefCovD : Computing RiemannToWeylRules for dim 4
** DefCovD : Computing RicciToTFRicci for dim 4
** DefCovD : Computing RicciToEinsteinRules for dim 4
** DefTensor : Defining weight +2 density DetMet []. Determinant .
```

Perturbations of the metric are written as δg.

In[7]:=
```
DefMetricPerturbation [Met, δMet, ϵ, PrintAs → "δg"]
```

```
** DefParameter : Defining parameter ϵ.
** DefTensor : Defining tensor δMet[LI[order], -α, -β].
```

Minkowski metric

We also define the Minkowski metric as a flat metric:

In[ ]:=
```
DefMetric[{3, 1, 0}, Mink[-μ, -ν], PD, {",", "∂"}, FlatMetric → True, PrintAs → "η"]
```

```
DefMetric : There are already metrics {Met} in vbundle TM.
** DefTensor : Defining symmetric metric tensor Mink[-μ, -ν].
** DefTensor : Defining inverse metric tensor InvMink[μ, ν]. Metric is frozen!
** DefTensor : Defining antisymmetric tensor epsilonMink [-α, -β, -γ, -ζ].
** DefTensor : Defining tetrametric TetraMink [-α, -β, -γ, -ζ].
** DefTensor : Defining tetrametric TetraMink† [-α, -β, -γ, -ζ].
** DefMetric : Associating fiducial flat derivative PD to metric .
** DefTensor : Defining weight +2 density DetMink []. Determinant .
```

Taylor series expansion

Definition

Before studying perturbations of the metric tensor and other tensor derived from the metric, it is helpful to briefly recall the Taylor expansion of a function f around some value x:

In[9]:=
```
Series[f[x + δx], {δx, 0, 3}]
```

Out[ ]= $f[x] + f'[x]\,\delta x + \frac{1}{2} f''[x]\,\delta x^2 + \frac{1}{6} f^{(3)}[x]\,\delta x^3 + O[\delta x]^4$

In the following, we will consider only the series up to the first term, which is linear in the perturbation δx of the argument:

In[ ]:=
```
Series[f[x + δx], {δx, 0, 1}]
```

Out[ ]= $f[x] + f'[x]\,\delta x + O[\delta x]^2$

This is the coefficient of this term

In[ ]:=
```
SeriesCoefficient [f[x + δx], {δx, 0, 1}]
```

Out[ ]= $f'[x]$

Product rule

Since the linear perturbation is given by the derivative, the product rule holds:

In[12]:=
```
SeriesCoefficient [f[x + δx] × g[x + δx], {δx, 0, 1}]
```
Out[ ]= $g[x]\, f'[x] + f[x]\, g'[x]$

Inverse rule

Similarly to the product rule, we have the rule

In[13]:=
```
SeriesCoefficient [1 / f[x + δx], {δx, 0, 1}]
```
Out[ ]= $-\frac{f'[x]}{f[x]^2}$

Notational conventions

Perturbation of the metric tensor

The metric is expanded in a formal Taylor series expansion, where the n'th perturbation order carries a factor $\frac{\epsilon^n}{n!}$ and is written as $\delta g^n_{\mu\nu}$. Here we will be interested only in the linear perturbations.

In[14]:=
```
Perturbed [Met[-μ, -ν], 3]
```
Out[ ]= $g_{\mu\nu} + \epsilon\ \delta g^1{}_{\mu\nu} + \frac{1}{2}\epsilon^2\ \delta g^2{}_{\mu\nu} + \frac{1}{6}\epsilon^3\ \delta g^3{}_{\mu\nu}$

This gives us a particular perturbation order.

In[15]:=
```
Perturbation [Met[-μ, -ν], 2]
```
Out[ ]= $\delta g^2{}_{\mu\nu}$

Perturbation of other tensors

For other tensors, the n'th order perturbation is written as $\triangle^n[X]$:

In[16]:=
```
Perturbation [RicciCD [-μ, -ν], 2]
```
Out[ ]= $\triangle^2[R[\nabla]_{\mu\nu}]$

The formal series expansion has the same form:

In[17]:=
```
Perturbed [RicciCD [-μ, -ν], 3]
```
Out[ ]= $\epsilon\ \triangle[R[\nabla]_{\mu\nu}] + \frac{1}{2}\epsilon^2\ \triangle^2[R[\nabla]_{\mu\nu}] + \frac{1}{6}\epsilon^3\ \triangle^3[R[\nabla]_{\mu\nu}] + R[\nabla]_{\mu\nu}$

Perturbations

Inverse metric

We start by calculating the perturbation $\triangle[g^{\mu\nu}]$ of inverse $g^{\mu\nu}$ of the metric tensor. This can be obtained from two steps. First, note the relation between the metric and its inverse, whose contraction yields a Kronecker symbol:

In[18]:=
```
Inactive[Met[μ, ρ]] × Inactive[Met[-ρ, -ν]]
Activate[%]
```

Out[ ]= $\text{Inactive}[g^{\mu\rho}] \times \text{Inactive}[g_{\rho\nu}]$

Out[ ]= $\delta_{\nu}{}^{\mu}$

Second, since the Kronecker symbol does not depend on the metric, its perturbation vanishes:

In[20]:=
```
Perturbation[delta[-μ, ν]]
```

Out[ ]= 0

We can use this to calculate the perturbation of the inverse metric. Here we use the product rule:

In[21]:=
```
Perturbation[Inactive[Met[μ, ρ]] × Inactive[Met[-ρ, -σ]]]
Activate[%]
Expand[Met[ν, σ] %]
Solve[% == 0, Perturbation[Met[μ, ν]]][[1]]
```

Out[ ]= $\text{Inactive}[g_{\rho\sigma}]\triangle[\text{Inactive}[g^{\mu\rho}]] + \text{Inactive}[g^{\mu\rho}]\triangle[\text{Inactive}[g_{\rho\sigma}]]$

Out[ ]= $g_{\rho\sigma}\triangle[g^{\mu\rho}] + g^{\mu\rho}\ \delta g^{1}{}_{\rho\sigma}$

Out[ ]= $\triangle[g^{\mu\nu}] + g^{\mu\rho}\ g^{\nu\sigma}\ \delta g^{1}{}_{\rho\sigma}$

Out[ ]= $\{\triangle[g^{\mu\nu}] \to -g^{\mu\rho}\ g^{\nu\sigma}\ \delta g^{1}{}_{\rho\sigma}\}$

We can also do so explicitly using the Taylor expansion in the perturbation parameter:

In[29]:=
```
Perturbed[Met[μ, ρ], 1] × Perturbed[Met[-ρ, -σ], 1]
SeriesCoefficient[%, {ϵ, 0, 1}]
Expand[Met[ν, σ] %]
Solve[% == 0, Perturbation[Met[μ, ν]]][[1]]
```

Out[ ]= $(g^{\mu\rho} + \epsilon\triangle[g^{\mu\rho}])(g_{\rho\sigma} + \epsilon\ \delta g^{1}{}_{\rho\sigma})$

Out[ ]= $g_{\rho\sigma}\triangle[g^{\mu\rho}] + g^{\mu\rho}\ \delta g^{1}{}_{\rho\sigma}$

Out[ ]= $\triangle[g^{\mu\nu}] + g^{\mu\rho}\ g^{\nu\sigma}\ \delta g^{1}{}_{\rho\sigma}$

Out[ ]= $\{\triangle[g^{\mu\nu}] \to -g^{\mu\rho}\ g^{\nu\sigma}\ \delta g^{1}{}_{\rho\sigma}\}$

In other words, the perturbation of the inverse metric is given by taking the negative of the perturbation of the metric, and then raising its indices. This confirms our result:

In[ ]:=
```
Perturbation[Met[μ, ν]]
ExpandPerturbation[%]
SeparateMetric[][%]
```

Out[ ]= $\Delta[g^{\mu\nu}]$

Out[ ]= $-\delta g^{1\mu\nu}$

Out[ ]= $-g^{\mu\alpha}\, g^{\nu\beta}\, \delta g^{1}{}_{\alpha\beta}$

Christoffel symbols

We then come to the perturbation of the Christoffel symbols. Their definition is well-known:

In[ ]:=
```
ChristoffelToGradMetric[ChristoffelCD[μ, -ρ, -ν]]
```

Out[ ]= $\frac{1}{2}\, g^{\mu\alpha}\, (-\partial_\alpha g_{\rho\nu} + \partial_\nu g_{\rho\alpha} + \partial_\rho g_{\nu\alpha})$

Since we have already calculated the perturbation of the inverse metric, we can use this formula. The perturbation of the derivative of the metric is just the derivative of the perturbation (partial derivative and perturbing commute):

In[ ]:=
```
Inactive[Perturbation[PD[-ρ][Met[-μ, -ν]]]]
Activate[%]
```

Out[ ]= $\text{Inactive}[\Delta[\partial_\rho g_{\mu\nu}]]$

Out[ ]= $\partial_\rho \delta g^{1}{}_{\mu\nu}$

We then use this rule to calculate the perturbation of the Christoffel symbols. It is helpful to realize that the partial derivatives which appear can be rewritten in terms of covariant derivatives, so that in the end all Christoffel symbols cancel:

In[39]:=
```
ChristoffelToGradMetric [ChristoffelCD [μ, -ρ, -ν]];
Perturbation [%];
ExpandPerturbation [%]
ChangeCovD [%, PD, CD];
Expand [%];
ContractMetric [%]
ToCanonical [%];
Simplify [%]
SeparateMetric [][%]
```

Out[ ]= $\frac{1}{2}\left(-\delta g^{1\mu\alpha}\left(-\partial_\alpha g_{\rho\nu}+\partial_\nu g_{\rho\alpha}+\partial_\rho g_{\nu\alpha}\right)+g^{\mu\alpha}\left(-\partial_\alpha \delta g^1{}_{\rho\nu}+\partial_\nu \delta g^1{}_{\rho\alpha}+\partial_\rho \delta g^1{}_{\nu\alpha}\right)\right)$

Out[ ]= $\frac{1}{2}\Gamma[\nabla]^\alpha{}_{\nu\rho}\ \delta g^1{}_\alpha{}^\mu+\frac{1}{2}\Gamma[\nabla]^\alpha{}_{\rho\nu}\ \delta g^1{}_\alpha{}^\mu-\frac{1}{2}\Gamma[\nabla]^{\alpha\mu}{}_\rho\ \delta g^1{}_{\alpha\nu}-\frac{1}{2}\Gamma[\nabla]^\alpha{}_{\nu\rho}\ \delta g^{1\mu}{}_\alpha-\frac{1}{2}\Gamma[\nabla]^\alpha{}_{\rho\nu}\ \delta g^{1\mu}{}_\alpha+\frac{1}{2}\Gamma[\nabla]_{\nu\alpha\rho}\ \delta g^{1\mu\alpha}-\frac{1}{2}\Gamma[\nabla]_{\nu\rho\alpha}\ \delta g^{1\mu\alpha}+\frac{1}{2}\Gamma[\nabla]_{\rho\alpha\nu}\ \delta g^{1\mu\alpha}-\frac{1}{2}\Gamma[\nabla]_{\rho\nu\alpha}\ \delta g^{1\mu\alpha}+\frac{1}{2}\Gamma[\nabla]^\alpha{}_\rho{}^\mu\ \delta g^1{}_{\nu\alpha}-\frac{1}{2}\Gamma[\nabla]^{\alpha\mu}{}_\nu\ \delta g^1{}_{\rho\alpha}+\frac{1}{2}\Gamma[\nabla]^\alpha{}_\nu{}^\mu\ \delta g^1{}_{\rho\alpha}-\frac{1}{2}\left(\nabla^\mu\ \delta g^1{}_{\rho\nu}\right)+\frac{1}{2}\left(\nabla_\nu\ \delta g^1{}_\rho{}^\mu\right)+\frac{1}{2}\left(\nabla_\rho\ \delta g^1{}_\nu{}^\mu\right)$

Out[ ]= $\frac{1}{2}\left(-\left(\nabla^\mu\ \delta g^1{}_{\nu\rho}\right)+\nabla_\nu\ \delta g^{1\mu}{}_\rho+\nabla_\rho\ \delta g^{1\mu}{}_\nu\right)$

Out[ ]= $\frac{1}{2}\left(-g^{\mu\alpha}\left(\nabla_\alpha\ \delta g^1{}_{\nu\rho}\right)+g^{\mu\beta}\left(\nabla_\nu\ \delta g^1{}_{\beta\rho}\right)+g^{\mu\gamma}\left(\nabla_\rho\ \delta g^1{}_{\gamma\nu}\right)\right)$

Hence, the perturbation of the Christoffel symbols is given as the covariant derivative of the metric perturbation. This shows us an important property which we will use later: the perturbation of the Christoffel symbols is a tensor. Again we check our result:

In[ ]:=
```
Perturbation [ChristoffelCD [μ, -ρ, -ν]]
ExpandPerturbation [%];
SeparateMetric [][%]
```

Out[ ]= $\Delta\left[\Gamma[\nabla]^\mu{}_{\rho\nu}\right]$

Out[ ]= $\frac{1}{2}\left(-g^{\mu\alpha}\left(\nabla_\alpha\ \delta g^1{}_{\rho\nu}\right)+g^{\mu\beta}\left(\nabla_\nu\ \delta g^1{}_{\beta\rho}\right)+g^{\mu\gamma}\left(\nabla_\rho\ \delta g^1{}_{\gamma\nu}\right)\right)$

Riemann curvature tensor

Next we expand the Riemann curvature tensor. It is defined as follows - note the order of the indices, which is reversed, compared to the most common convention:

In[51]:=
```
RiemannCD [-σ, -ρ, -ν, μ]
ChangeCurvature [%]
```

Out[ ]= $R[\nabla]_{\sigma\rho\nu}{}^\mu$

Out[ ]= $\Gamma[\nabla]^\alpha{}_{\sigma\nu}\ \Gamma[\nabla]^\mu{}_{\rho\alpha}-\Gamma[\nabla]^\alpha{}_{\rho\nu}\ \Gamma[\nabla]^\mu{}_{\sigma\alpha}+\partial_\rho\Gamma[\nabla]^\mu{}_{\sigma\nu}-\partial_\sigma\Gamma[\nabla]^\mu{}_{\rho\nu}$

We can use this formula to calculate its perturbation. Now we use our previously gained knowledge that the perturbation of the Christoffel symbol is a tensor. We can thus combine partial derivatives and Christoffel symbols to covariant derivatives to get the desired result:

In[53]:=
```
RiemannCD[-σ, -ρ, -ν, μ];
ChangeCurvature[%];
Perturbation[%]
ChangeCovD[%, PD, CD];
ToCanonical[%]
ExpandPerturbation[%];
ToCanonical[%];
Simplify[%]
```

Out[ ]= $-\Gamma[\nabla]^{\mu}{}_{\sigma\alpha}\,\Delta[\Gamma[\nabla]^{\alpha}{}_{\rho\nu}] + \Gamma[\nabla]^{\mu}{}_{\rho\alpha}\,\Delta[\Gamma[\nabla]^{\alpha}{}_{\sigma\nu}] + \Gamma[\nabla]^{\alpha}{}_{\sigma\nu}\,\Delta[\Gamma[\nabla]^{\mu}{}_{\rho\alpha}] - \Gamma[\nabla]^{\alpha}{}_{\rho\nu}\,\Delta[\Gamma[\nabla]^{\mu}{}_{\sigma\alpha}] + \partial_{\rho}\Delta[\Gamma[\nabla]^{\mu}{}_{\sigma\nu}] - \partial_{\sigma}\Delta[\Gamma[\nabla]^{\mu}{}_{\rho\nu}]$

Out[ ]= $\nabla_{\rho}\,\Delta[\Gamma[\nabla]^{\mu}{}_{\nu\sigma}] - \nabla_{\sigma}\,\Delta[\Gamma[\nabla]^{\mu}{}_{\nu\rho}]$

Out[ ]= $\frac{1}{2}\left(-(\nabla_{\rho}\nabla^{\mu}\,\delta g^{1}{}_{\nu\sigma}) + \nabla_{\rho}\nabla_{\nu}\,\delta g^{1\mu}{}_{\sigma} + \nabla_{\rho}\nabla_{\sigma}\,\delta g^{1\mu}{}_{\nu} + \nabla_{\sigma}\nabla^{\mu}\,\delta g^{1}{}_{\nu\rho} - \nabla_{\sigma}\nabla_{\nu}\,\delta g^{1\mu}{}_{\rho} - \nabla_{\sigma}\nabla_{\rho}\,\delta g^{1\mu}{}_{\nu}\right)$

We see that the perturbation is given by the second covariant derivative of the metric tensor. This confirms our result:

In[ ]:=
```
Perturbation[RiemannCD[-σ, -ρ, -ν, μ]]
ExpandPerturbation[%];
Expand[%];
ToCanonical[%];
Simplify[%]
```

Out[ ]= $\Delta[R[\nabla]_{\sigma\rho\nu}{}^{\mu}]$

Out[ ]= $\frac{1}{2}\left(-(\nabla_{\rho}\nabla^{\mu}\,\delta g^{1}{}_{\nu\sigma}) + \nabla_{\rho}\nabla_{\nu}\,\delta g^{1\mu}{}_{\sigma} + \nabla_{\rho}\nabla_{\sigma}\,\delta g^{1\mu}{}_{\nu} + \nabla_{\sigma}\nabla^{\mu}\,\delta g^{1}{}_{\nu\rho} - \nabla_{\sigma}\nabla_{\nu}\,\delta g^{1\mu}{}_{\rho} - \nabla_{\sigma}\nabla_{\rho}\,\delta g^{1\mu}{}_{\nu}\right)$

Ricci tensor

We continue with the Ricci tensor, which is obtained by contraction of two indices of the curvature tensor:

In[ ]:=
```
Inactive[RiemannCD[-μ, -ρ, -ν, ρ]]
Activate[%]
```

Out[ ]= Inactive[$R[\nabla]_{\mu\rho\nu}{}^{\rho}$]

Out[ ]= $R[\nabla]_{\mu\nu}$

Contracting two indices on the perturbation of the curvature tensor therefore gives us the desired result:

In[68]:=
```
Perturbation[RiemannCD[-μ, -ρ, -ν, σ]]
ExpandPerturbation[%];
Expand[%];
ToCanonical[%];
Simplify[%]
delta[-σ, ρ] %
Expand[%];
ToCanonical[%];
Simplify[%]
```

Out[ ]= $\triangle[R[\nabla]_{\mu\rho\nu}{}^{\sigma}]$

Out[ ]= $\frac{1}{2}\left(-\left(\nabla_\mu \nabla_\nu \delta g^1{}_\rho{}^\sigma\right) - \nabla_\mu \nabla_\rho \delta g^1{}_\nu{}^\sigma + \nabla_\mu \nabla^\sigma \delta g^1{}_{\nu\rho} + \nabla_\rho \nabla_\mu \delta g^1{}_\nu{}^\sigma + \nabla_\rho \nabla_\nu \delta g^1{}_\mu{}^\sigma - \nabla_\rho \nabla^\sigma \delta g^1{}_{\mu\nu}\right)$

Out[ ]= $\frac{1}{2}\left(-\left(\nabla_\mu \nabla_\nu \delta g^1{}_\sigma{}^\sigma\right) - \nabla_\mu \nabla_\sigma \delta g^1{}_\nu{}^\sigma + \nabla_\mu \nabla^\sigma \delta g^1{}_{\nu\sigma} + \nabla_\sigma \nabla_\mu \delta g^1{}_\nu{}^\sigma + \nabla_\sigma \nabla_\nu \delta g^1{}_\mu{}^\sigma - \nabla_\sigma \nabla^\sigma \delta g^1{}_{\mu\nu}\right)$

Out[ ]= $\frac{1}{2}\left(-\left(\nabla_\nu \nabla_\mu \delta g^{1\sigma}{}_\sigma\right) + \nabla_\sigma \nabla_\mu \delta g^1{}_\nu{}^\sigma + \nabla_\sigma \nabla_\nu \delta g^1{}_\mu{}^\sigma - \nabla_\sigma \nabla^\sigma \delta g^1{}_{\mu\nu}\right)$

Here two identical terms cancel. This confirms our result:

In[ ]:=
```
Perturbation[RicciCD[-μ, -ν]]
ExpandPerturbation[%];
Expand[%];
ToCanonical[%];
Simplify[%]
```

Out[ ]= $\triangle[R[\nabla]_{\mu\nu}]$

Out[ ]= $\frac{1}{2}\left(-\left(\nabla_\alpha \nabla^\alpha \delta g^1{}_{\mu\nu}\right) + \nabla_\alpha \nabla_\mu \delta g^1{}_\nu{}^\alpha + \nabla_\alpha \nabla_\nu \delta g^1{}_\mu{}^\alpha - \nabla_\nu \nabla_\mu \delta g^{1\alpha}{}_\alpha\right)$

Ricci scalar

The Ricci scalar is obtained by contracting the Ricci tensor with the metric:

In[ ]:=
```
RicciCD[-μ, -ν] × Met[μ, ν]
```

Out[ ]= $R[\nabla]$

Again we can make use of our previous knowledge and the product rule:

In[ ]:=
```
Perturbation[Inactive[RicciCD[-μ, -ν]] × Inactive[Met[μ, ν]]];
Activate[%]
ExpandPerturbation[%];
Expand[%];
ToCanonical[%];
Simplify[%]
```

Out[ ]= $g^{\mu\nu}\ \Delta[R[\nabla]_{\mu\nu}] + \Delta[g^{\mu\nu}]\ R[\nabla]_{\mu\nu}$

Out[ ]= $-R[\nabla]^{\alpha\mu}\ \delta g^{1}{}_{\alpha\mu} - \frac{1}{2}\ g^{\alpha\mu}\ (\nabla_{\mu}\nabla_{\alpha}\delta g^{1\nu}{}_{\nu} - 2\,(\nabla_{\nu}\nabla_{\mu}\delta g^{1}{}_{\alpha}{}^{\nu}) + \nabla_{\nu}\nabla^{\nu}\delta g^{1}{}_{\alpha\mu})$

This can also be obtained directly:

In[ ]:=
```
Perturbation[RicciScalarCD[]]
ExpandPerturbation[%];
Expand[%];
ContractMetric[%];
ToCanonical[%];
Simplify[%]
```

Out[ ]= $\Delta[R[\nabla]]$

Out[ ]= $-R[\nabla]^{\alpha\beta}\ \delta g^{1}{}_{\alpha\beta} + \nabla_{\beta}\nabla_{\alpha}\delta g^{1\alpha\beta} - \nabla_{\beta}\nabla^{\beta}\delta g^{1\alpha}{}_{\alpha}$

Einstein tensor

Finally, we calculate the perturbation of the Einstein tensor. Its definition is well-known, and so we obtains its perturbation:

In[ ]:=
```
EinsteinToRicci[EinsteinCD[-μ, -ν]]
Perturbation[%]
ExpandPerturbation[%];
Expand[%];
ContractMetric[%];
ToCanonical[%];
Simplify[%]
```

Out[ ]= $R[\nabla]_{\mu\nu} - \frac{1}{2}\ g_{\nu\mu}\ R[\nabla]$

Out[ ]= $\Delta[R[\nabla]_{\mu\nu}] + \frac{1}{2}\,(-\ g_{\nu\mu}\ \Delta[R[\nabla]] - R[\nabla]\ \delta g^{1}{}_{\nu\mu})$

Out[ ]= $\frac{1}{2}\,(-R[\nabla]\ \delta g^{1}{}_{\mu\nu} - \nabla_{\alpha}\nabla^{\alpha}\delta g^{1}{}_{\mu\nu} + \nabla_{\alpha}\nabla_{\mu}\delta g^{1}{}_{\nu}{}^{\alpha} + \nabla_{\alpha}\nabla_{\nu}\delta g^{1}{}_{\mu}{}^{\alpha} + g_{\mu\nu}\ (R[\nabla]^{\alpha\beta}\ \delta g^{1}{}_{\alpha\beta} - \nabla_{\beta}\nabla_{\alpha}\delta g^{1\alpha\beta} + \nabla_{\beta}\nabla^{\beta}\delta g^{1\alpha}{}_{\alpha}) - \nabla_{\nu}\nabla_{\mu}\delta g^{1\alpha}{}_{\alpha})$

We see that it is significantly more complicated than the perturbation of the Ricci tensor; hence, whenever we consider the perturbation of the vacuum Einstein equations, it is much simpler to consider their trace-reversed form $R_{\mu\nu} = 0$ instead of $G_{\mu\nu} = 0$. We finally check:

In[ ]:=
```
Perturbation[EinsteinCD[-μ, -ν]]
ExpandPerturbation[%];
Expand[%];
ContractMetric[%];
ToCanonical[%];
Simplify[%]
```

Out[ ]= $\triangle[G[\nabla]_{\mu\nu}]$

Out[ ]= $\frac{1}{2}\left(-R[\nabla]\ \delta g^{1}{}_{\mu\nu} - \nabla_{\alpha}\nabla^{\alpha}\ \delta g^{1}{}_{\mu\nu} + \nabla_{\alpha}\nabla_{\mu}\ \delta g^{1}{}_{\nu}{}^{\alpha} + \nabla_{\alpha}\nabla_{\nu}\ \delta g^{1}{}_{\mu}{}^{\alpha} + g_{\mu\nu}\ \left(R[\nabla]^{\alpha\beta}\ \delta g^{1}{}_{\alpha\beta} - \nabla_{\beta}\nabla_{\alpha}\ \delta g^{1\alpha\beta} + \nabla_{\beta}\nabla^{\beta}\ \delta g^{1\alpha}{}_{\alpha}\right) - \nabla_{\nu}\nabla_{\mu}\ \delta g^{1\alpha}{}_{\alpha}\right)$

Perturbations around the Minkowski metric

Collecting perturbations

Recall from the previous section the objects we studied and their perturbations:

In[ ]:=
```
tens = {Met[-μ, -ν], Met[μ, ν], ChristoffelCD[μ, -ρ, -ν],
  RiemannCD[-σ, -ρ, -ν, μ], RicciCD[-μ, -ν], RicciScalarCD[], EinsteinCD[-μ, -ν]}
Perturbation /@ %;
ExpandPerturbation /@ %;
Expand /@ %;
ToCanonical /@ %;
pert = Simplify /@ %;
```

Out[ ]= $\{g_{\mu\nu},\ g^{\mu\nu},\ \Gamma[\nabla]^{\mu}{}_{\rho\nu},\ R[\nabla]_{\sigma\rho\nu}{}^{\mu},\ R[\nabla]_{\mu\nu},\ R[\nabla],\ G[\nabla]_{\mu\nu}\}$

Setting background to Minkowski

In the following, we will set the metric to the Minkowski metric, and likewise for the inverse:

In[ ]:=
```
gtoη =
 Join[MakeRule[{Met[-μ, -ν], Mink[-μ, -ν]}, MetricOn → None, ContractMetrics → False],
  MakeRule[{Met[μ, ν], InvMink[μ, ν]}, MetricOn → None, ContractMetrics → False]]
```

Out[ ]= $\{\text{HoldPattern}[g_{\underline{\mu}\underline{\nu}}] :\to \text{Module}[\{\},\ \eta_{\mu\nu}],\ \text{HoldPattern}[g^{\underline{\mu}\underline{\nu}}] :\to \text{Module}[\{\},\ i\eta^{\mu\nu}]\}$

To translate the perturbations from a general background to a Minkowski background, we perform the following steps:

- Bring all indices to their natural position, insert metrics or inverse metrics as needed.
- Change all curvature terms to derivatives and products of Christoffel symbols.
- Change all covariant derivatives to partial derivatives and Christoffel symbols.
- Change all Christoffel symbols to derivatives of the metric.

- Replace the metric and its inverse by the Minkowski metric.

```
pert;
SeparateMetric[] /@ %;
ChangeCurvature /@ %;
ChangeCovD[#, CD, PD] & /@ %;
ChristoffelToGradMetric /@ %;
% //. gtoη;
Expand /@ %;
ContractMetric /@ %;
ToCanonical[#, UseMetricOnVBundle → None] & /@ %;
flat = Simplify /@ %;
```

Comparison of results

The following table compares the perturbations around a general background and around the Minkowski metric:

```
Grid[Prepend[ScreenDollarIndices[Transpose[{tens, pert, flat}]],
  {"Object", "General background metric", "Minkowski background"}], Frame → All]
```

Object	General background metric	Minkowski background
$g_{\mu\nu}$	$\delta g^1{}_{\mu\nu}$	$\delta g^1{}_{\mu\nu}$
$g^{\mu\nu}$	$-\delta g^{1\mu\nu}$	$-i\eta^{\mu\alpha}\ i\eta^{\nu\beta}\ \delta g^1{}_{\alpha\beta}$
$\Gamma[\nabla]^{\mu}{}_{\rho\nu}$	$\frac{1}{2}(-(\nabla^\mu \delta g^1{}_{\nu\rho}) + \nabla_\nu \delta g^{1\mu}{}_\rho + \nabla_\rho \delta g^{1\mu}{}_\nu)$	$\frac{1}{2}\ i\eta^{\mu\alpha}\ (-\partial_\alpha \delta g^1{}_{\nu\rho} + \partial_\nu \delta g^1{}_{\rho\alpha} + \partial_\rho \delta g^1{}_{\nu\alpha})$
$R[\nabla]_{\sigma\rho\nu}{}^{\mu}$	$\frac{1}{2}(-(\nabla_\rho \nabla^\mu \delta g^1{}_{\nu\sigma}) + \nabla_\rho \nabla_\nu \delta g^{1\mu}{}_\sigma + \nabla_\rho \nabla_\sigma \delta g^{1\mu}{}_\nu + \nabla_\sigma \nabla^\mu \delta g^1{}_{\nu\rho} - \nabla_\sigma \nabla_\nu \delta g^{1\mu}{}_\rho - \nabla_\sigma \nabla_\rho \delta g^{1\mu}{}_\nu)$	$\frac{1}{2}\ i\eta^{\mu\alpha}\ (-\partial_\alpha\partial_\rho\delta g^1{}_{\nu\sigma} + \partial_\alpha\partial_\sigma\delta g^1{}_{\nu\rho} + \partial_\rho\partial_\nu\delta g^1{}_{\sigma\alpha} - \partial_\sigma\partial_\nu\delta g^1{}_{\rho\alpha})$
$R[\nabla]_{\mu\nu}$	$\frac{1}{2}(-(\nabla_\alpha \nabla^\alpha \delta g^1{}_{\mu\nu}) + \nabla_\alpha \nabla_\mu \delta g^1{}_\nu{}^\alpha + \nabla_\alpha \nabla_\nu \delta g^1{}_\mu{}^\alpha - \nabla_\nu \nabla_\mu \delta g^{1\alpha}{}_\alpha)$	$\frac{1}{2}\ i\eta^{\alpha\beta}\ (-\partial_\beta\partial_\alpha\delta g^1{}_{\mu\nu} + \partial_\beta\partial_\mu\delta g^1{}_{\nu\alpha} + \partial_\beta\partial_\nu\delta g^1{}_{\mu\alpha} - \partial_\nu\partial_\mu\delta g^1{}_{\alpha\beta})$
$R[\nabla]$	$-R[\nabla]^{\alpha\beta}\ \delta g^1{}_{\alpha\beta} - \frac{1}{2}\ g^{\alpha\beta}\ (\nabla_\beta \nabla_\alpha \delta g^{1\gamma}{}_\gamma - 2(\nabla_\gamma \nabla_\beta \delta g^1{}_\alpha{}^\gamma) + \nabla_\gamma \nabla^\gamma \delta g^1{}_{\alpha\beta})$	$i\eta^{\alpha\beta}\ i\eta^{\gamma\zeta}\ (\partial_\zeta\partial_\beta\delta g^1{}_{\alpha\gamma} - \partial_\zeta\partial_\gamma\delta g^1{}_{\alpha\beta})$
$G[\nabla]_{\mu\nu}$	$\frac{1}{4}(g_{\mu\nu}\ (2\ R[\nabla]^{\alpha\beta}\ \delta g^1{}_{\alpha\beta} + g^{\alpha\beta}\ (\nabla_\beta \nabla_\alpha \delta g^{1\gamma}{}_\gamma - 2(\nabla_\gamma \nabla_\beta \delta g^1{}_\alpha{}^\gamma) + \nabla_\gamma \nabla^\gamma \delta g^1{}_{\alpha\beta})) - 2(R[\nabla]\ \delta g^1{}_{\mu\nu} + \nabla_\alpha \nabla^\alpha \delta g^1{}_{\mu\nu} - \nabla_\alpha \nabla_\mu \delta g^1{}_\nu{}^\alpha - \nabla_\alpha \nabla_\nu \delta g^1{}_\mu{}^\alpha + \nabla_\nu \nabla_\mu \delta g^{1\alpha}{}_\alpha))$	$\frac{1}{2}\ i\eta^{\alpha\beta}\ (-\partial_\beta\partial_\alpha\delta g^1{}_{\mu\nu} + \partial_\beta\partial_\mu\delta g^1{}_{\nu\alpha} + \partial_\beta\partial_\nu\delta g^1{}_{\mu\alpha} - i\eta^{\gamma\zeta}\ \eta_{\mu\nu}\ \partial_\zeta\partial_\beta\delta g^1{}_{\alpha\gamma} + i\eta^{\gamma\zeta}\ \eta_{\mu\nu}\ \partial_\zeta\partial_\gamma\delta g^1{}_{\alpha\beta} - \partial_\nu\partial_\mu\delta g^1{}_{\alpha\beta})$